



English

The Fast Track to julia^{1.0}

A quick and dirty overview of

This page's source is [located here](#) . Pull requests are welcome!

What is...?

Julia is an open-source, multi-platform, high-level, high-performance programming language for technical computing.

Julia has an LLVM-based JIT compiler that allows it to match the performance of languages such as C and FORTRAN without the hassle of low-level code. Because the code is compiled on the fly you can run (bits of) code in a shell or REPL , which is part of the recommended workflow .

Julia is dynamically typed, provides multiple dispatch , and is designed for parallelism and distributed computation.

Julia has a built-in package manager.

Julia has many built-in mathematical functions, including special functions (e.g. Gamma), and supports complex numbers right out of the box.

Julia allows you to generate code automatically thanks to Lisp-inspired macros.

Julia was created in 2012.

Basics

Assignment

```
answer = 42
x, y, z = 1, [1:10; ], "A string"
x, y = y, x # swap x and y
```

Constant declaration

```
const DATE_OF_BIRTH = 2012
```

End-of-line comment

```
i = 1 # This is a comment
```

Delimited comment

```
#= This is another comment =#
```

Chaining

```
x = y = z = 1 # right-to-left
0 < x < 3      # true
5 < x != y < 5 # false
```

Function definition

```
function add_one(i)
    return i + 1
end
```

Insert LaTeX symbols

```
\delta + [Tab]
```

Operators

Basic arithmetic	<code>+, -, *, /</code>
Exponentiation	<code>2^3 == 8</code>
Division	<code>3/12 == 0.25</code>
Inverse division	<code>7\3 == 3/7</code>
Remainder	<code>x % y</code> or <code>rem(x,y)</code>
Negation	<code>!true == false</code>
Equality	<code>a == b</code>
Inequality	<code>a != b</code> or <code>a ≠ b</code>
Less and larger than	<code><</code> and <code>></code>
Less than or equal to	<code><=</code> or <code>≤</code>
Greater than or equal to	<code>>=</code> or <code>≥</code>
Element-wise operation	<code>[1, 2, 3] .+ [1, 2, 3] == [2, 4, 6]</code> <code>[1, 2, 3] .* [1, 2, 3] == [1, 4, 9]</code>
Not a number	<code>isnan(NaN)</code> <code>not(!) NaN == NaN</code>
Ternary operator	<code>a == b ? "Equal" : "Not equal"</code>
Short-circuited AND and OR	<code>a && b</code> and <code>a b</code>
Object equivalence	<code>a === b</code>

The shell a.k.a. REPL

Recall last result	<code>ans</code>
Interrupt execution	<code>[Ctrl] + [C]</code>
Clear screen	<code>[Ctrl] + [L]</code>
Run program	<code>include("filename.jl")</code>
Get help for <code>func</code> is defined	<code>?func</code>
See all places where <code>func</code> is defined	<code>apropos("func")</code>
Command line mode	<code>;</code> on empty line
Package Manager mode	<code>]</code> on empty line
Help mode	<code>?</code> on empty line
Exit special mode / Return to REPL	<code>[Backspace]</code> on empty line
Exit REPL	<code>exit()</code> or <code>[Ctrl] + [D]</code>

Standard libraries

To help Julia load faster, many core functionalities exist in standard libraries that come bundled with Julia. To make their functions available, use `using PackageName`. Here are some Standard Libraries and popular functions.

Random	<code>rand, randn, randsubseq</code>
Statistics	<code>mean, std, cor, median, quantile</code>
LinearAlgebra	<code>I, eigvals, eigvecs, det, cholesky</code>
SparseArrays	<code>sparse, SparseVector, SparseMatrixCSC</code>
Distributed	<code>@distributed, pmap, addprocs</code>
Dates	<code>DateTime, Date</code>

Package management

Packages must be **registered** before they are visible to the package manager. In Julia 1.0, there are two ways to work with the package manager: either with `using Pkg` and using `Pkg` functions, or by typing `]` in the REPL to enter the special interactive package management mode. (To return to regular REPL, just hit `BACKSPACE` on an empty line in package management mode). Note that new tools arrive in interactive mode first, then usually also become available in regular Julia sessions through `Pkg` module.

Using `Pkg` in Julia session

List installed packages (human-readable)	<code>Pkg.status()</code>
Update all packages	<code>Pkg.update()</code>
Install <code>PackageName</code>	<code>Pkg.add("PackageName")</code>
Rebuild <code>PackageName</code>	<code>Pkg.build("PackageName")</code>
Use <code>PackageName</code> (after install)	<code>using PackageName</code>
Remove <code>PackageName</code>	<code>Pkg.rm("PackageName")</code>

In Interactive Package Mode

Add <code>PackageName</code>	<code>add PackageName</code>
Remove <code>PackageName</code>	<code>rm PackageName</code>
Update <code>PackageName</code>	<code>update PackageName</code>
Use development version	<code>dev PackageName</code> or <code>dev GitRepoUrl</code>
Stop using development version, revert to public release	<code>free PackageName</code>

Characters and strings

Character	<code>chr = 'C'</code>
String	<code>str = "A string"</code>
Character code	<code>Int('J') == 74</code>
Character from code	<code>Char(74) == 'J'</code>
Any UTF character	<code>chr = '\uXXXX' # 4-digit HEX</code> <code>chr = '\UXXXXXXXX' # 8-digit HEX</code>
Loop through characters	<code>for c in str</code> <code>println(c)</code> <code>end</code>
Concatenation	<code>str = "Learn" * " " * "Julia"</code>
String interpolation	<code>a = b = 2</code> <code>println("a * b = \$(a*b)")</code>
First matching character or regular expression	<code>findfirst(isequal('i'), "Julia")</code> <code>== 4</code>
Replace substring or regular expression	<code>replace("Julia", "a" => "us") ==</code> <code>"Julius"</code>
Last index (of collection)	<code>lastindex("Hello") == 5</code>
Number of characters	<code>length("Hello") == 5</code>
Regular expression	<code>pattern = r"[aeiou]"</code> <code>str = "+1 234 567 890"</code> <code>pat = r"\+([0-9]) ([0-9]+)"</code> <code>m = match(pat, str)</code> <code>m.captures == ["1", "234"]</code> <code>[m.match for m = eachmatch(pat,</code> <code>str)]</code>
Subexpressions	<code>eachmatch(pat, str)</code>
All occurrences	
All occurrences (as iterator)	
Beware of multi-byte Unicode encodings in UTF-8:	
	<code>10 == lastindex("Ångström") != length("Ångström") == 8</code>
Strings are immutable.	

Numbers

Integer types	IntN and UIntN, with $N \in \{8, 16, 32, 64, 128\}$, BigInt
Floating-point types	FloatN with $N \in \{16, 32, 64\}$ BigFloat
Minimum and maximum values by type	typemin(Int8) typemax(Int64)
Complex types	Complex{T}
Imaginary unit	im
Machine precision	eps() # same as eps(Float64)
Rounding	round() # floating-point round(Int, x) # integer convert(TypeName, val) # attempt/error typename(val) # calls convert
Type conversions	pi # 3.1415... n # 3.1415... im # real(im * im) == -1
Global constants	
More constants	using Base.MathConstants
Julia does not automatically check for numerical overflow. Use package SaferIntegers for ints with overflow checking.	

Random Numbers

Many random number functions require using Random.

Set seed	seed!(seed)
Random numbers	rand() # uniform [0,1) randn() # normal (-Inf, Inf)
Random from Other Distribution	using Distributions my_dist = Bernoulli(0.2) # For example rand(my_dist)
Random subsample elements from A with inclusion probability p	randsubseq(A, p)
Random permutation elements of A	shuffle(A)

Arrays

Declaration	<code>arr = Float64[]</code>
Pre-allocation	<code>sizehint!(arr, 10^4)</code>
Access and assignment	<code>arr = Any[1,2]</code> <code>arr[1] = "Some text"</code>
Comparison	<code>a = [1:10;]</code> <code>b = a</code> <code># b points to a</code> <code>a[1] = -99</code> <code>a == b</code> <code># true</code>
Copy elements (not address)	<code>b = copy(a)</code> <code>b = deepcopy(a)</code>
Select subarray from m to n	<code>arr[m:n]</code>
n-element array with 0.0s	<code>zeros(n)</code>
n-element array with 1.0s	<code>ones(n)</code>
n-element array with #undefs	<code>Vector{Type}(undef,n)</code>
n equally spaced numbers from start to stop	<code>range(start,stop=stop,length=n)</code>
Array with n random Int8 elements	<code>rand{Int8, n}</code>
Fill array with val	<code>fill!(arr, val)</code>
Pop last element	<code>pop!(arr)</code>
Pop first element	<code>popfirst!(a)</code>
Push val as last element	<code>push!(arr, val)</code>
Push val as first element	<code>pushfirst!(arr, val)</code>
Remove element at index idx	<code>deleteat!(arr, idx)</code>
Sort	<code>sort!(arr)</code>
Append a with b	<code>append!(a,b)</code>
Check whether val is element	<code>in(val, arr)</code> or <code>val in arr</code>
Scalar product	<code>dot(a, b) == sum(a .* b)</code>
Change dimensions (if possible)	<code>reshape(1:6, 3, 2)' == [1 2 3; 4 5 6]</code>
To string (with delimiter del between elements)	<code>join(arr, del)</code>

Linear Algebra

For most linear algebra tools, use `using LinearAlgebra`.

Identity matrix	<code>I</code> # just use variable <code>I</code> . Will automatically conform to dimensions required.
Define matrix	<code>M = [1 0; 0 1]</code>
Matrix dimensions	<code>size(M)</code>
Select <code>i</code> th row	<code>M[i, :]</code>
Select <code>i</code> th column	<code>M[:, i]</code>
Concatenate horizontally	<code>M = [a b]</code> or <code>M = hcat(a, b)</code>
Concatenate vertically	<code>M = [a ; b]</code> or <code>M = vcat(a, b)</code>
Matrix transposition	<code>transpose(M)</code>
Conjugate matrix transposition	<code>M'</code> or <code>adjoint(M)</code>
Matrix trace	<code>tr(M)</code>
Matrix determinant	<code>det(M)</code>
Matrix rank	<code>rank(M)</code>
Matrix eigenvalues	<code>eigvals(M)</code>
Matrix eigenvectors	<code>eigvecs(M)</code>
Matrix inverse	<code>inv(M)</code>
Solve <code>M*x == v</code>	<code>M\v</code> is better than <code>inv(M)*v</code>
Moore-Penrose pseudo-inverse	<code>pinv(M)</code>

Julia has built-in support for [matrix decompositions](#).

Julia tries to infer whether matrices are of a special type (symmetric, hermitian, etc.), but sometimes fails. To aid Julia in dispatching the optimal algorithms, special matrices can be declared to have a structure with functions like `Symmetric`, `Hermitian`, `UpperTriangular`, `LowerTriangular`, `Diagonal`, and more.

Control flow and loops

Conditional	<code>if-elseif-else-end</code>
Simple for loop	<code>for i in 1:10 println(i) end</code>
Unnested for loop	<code>for i in 1:10, j = 1:5 println(i*j) end</code>
Enumeration	<code>for (idx, val) in enumerate(arr) println("the \$idx-th element is \$val") end</code>
while loop	<code>while bool_expr # do stuff end</code>
Exit loop	<code>break</code>
Exit iteration	<code>continue</code>

Functions

All arguments to functions are passed by reference.

Functions with `!` appended change at least one argument, typically the first: `sort!(arr)`.

Required arguments are separated with a comma and use the positional notation.

Optional arguments need a default value in the signature, defined with `=`.

Keyword arguments use the named notation and are listed in the function's signature after the semicolon:

```
function func(req1, req2; key1=df1t1, key2=df1t2)
    # do stuff
end
```

The semicolon is *not* required in the call to a function that accepts keyword arguments.

The `return` statement is optional but highly recommended.

Multiple data structures can be returned as a tuple in a single `return` statement.

Command line arguments `julia script.jl arg1 arg2...` can be processed from global constant `ARGS`:

```
for arg in ARGS
    println(arg)
end
```

Anonymous functions can best be used in collection functions or list comprehensions: `x -> x^2`.

Functions can accept a variable number of arguments:

```
function func(a...)
    println(a)
end
```

```
func(1, 2, [3:5]) # tuple: (1, 2, UnitRange{Int64}{3:5})
```


Functions can be nested:

```
function outerfunction()
    # do some outer stuff
    function innerfunction()
        # do inner stuff
        # can access prior outer definitions
    end
    # do more outer stuff
end
```

Functions can have explicit return types

```
# take any Number subtype and return it as a String
function stringifynumber(num::T)::String where T <: Number
    return "$num"
end
```

Functions can be **vectorized** by using the Dot Syntax

```
# here we broadcast the subtraction of each mean value
# by using the dot operator
julia> using Statistics
julia> A = rand(3, 4);
julia> B = A .- mean(A, dims=1)
3×4 Array{Float64,2}:
 0.0387438  0.112224 -0.0541478  0.455245
 0.000773337 0.250006  0.0140011 -0.289532
-0.0395171 -0.36223  0.0401467 -0.165713
julia> mean(B, dims=1)
1×4 Array{Float64,2}:
-7.40149e-17  7.40149e-17  1.85037e-17  3.70074e-17
```

Julia generates **specialized versions** of functions based on data types. When a function is called with the same argument types again, Julia can look up the native machine code and skip the compilation process.

Since **Julia 0.5** the existence of potential ambiguities is still acceptable, but actually calling an ambiguous method is an **immediate error**.

Stack overflow is possible when recursive functions nest many levels deep. **Trampolining** can be used to do tail-call optimization, as Julia does not do that automatically **yet**. Alternatively, you can rewrite the tail recursion as an iteration.

Dictionaries

Dictionary	<pre>d = Dict{key1 => val1, key2 => val2, ...} d = Dict{key1 => val1, :key2 => val2, ...}</pre>
All keys (iterator)	<code>keys(d)</code>
All values (iterator)	<code>values(d)</code>
Loop through key-value pairs	<pre>for (k,v) in d println("key: \$k, value: \$v") end</pre>
Check for key <code>:k</code>	<code>haskey(d, :k)</code>
Copy keys (or values) to array	<pre>arr = collect(keys(d)) arr = [k for (k,v) in d]</pre>
Dictionaries are mutable; when symbols are used as keys, the keys are immutable.	

Sets

Declaration	<code>s = Set([1, 2, 3, "Some text"])</code>
Union <code>s1 ∪ s2</code>	<code>union(s1, s2)</code>
Intersection <code>s1 ∩ s2</code>	<code>intersect(s1, s2)</code>
Difference <code>s1 \ s2</code>	<code>setdiff(s1, s2)</code>
Difference <code>s1 △ s2</code>	<code>symdiff(s1, s2)</code>
Subset <code>s1 ⊆ s2</code>	<code>issubset(s1, s2)</code>
Checking whether an element is contained in a set is done in O(1).	

Collection functions

Apply <code>f</code> to all elements of collection <code>coll</code>	<pre>map(f, coll) or map(coll) do elem # do stuff with elem # must contain return end</pre>
Filter <code>coll</code> for true values of <code>f</code>	<code>filter(f, coll)</code>
List comprehension	<code>arr = [f(elem) for elem in coll]</code>

Types

Julia has no classes and thus no class-specific methods.
Types are like classes without methods.
Abstract types can be subtyped but not instantiated.
Concrete types cannot be subtyped.
By default, `struct`s are immutable.

By default, `structs` are immutable.

Immutable types enhance performance and are thread safe, as they can be shared among threads without the need for synchronization.

Objects that may be one of a set of types are called `Union` types.

Type annotation	<code>var::TypeName</code>
Type declaration	<code>struct Programmer name::String birth_year::UInt16 fave_language::AbstractString end</code>
Mutable type declaration	replace struct with mutable struct
Type alias	<code>const Nerd = Programmer</code>
Type constructors	<code>methods(Typename)</code>
Type instantiation	<code>me = Programmer("Ian", 1984, "Julia") me = Nerd("Ian", 1984, "Julia")</code>
Subtype declaration	<code>abstract type Bird end struct Duck <: Bird pond::String end struct Point{T <: Real} x::T y::T end</code>
Parametric type	<code>p =Point{Float64}(1,2)</code>
Union types	<code>Union{Int, String}</code>
Traverse type hierarchy	<code>supertype(Typename)</code> and <code>subtypes(Typename)</code>
Default supertype	<code>Any</code>
All fields	<code>fieldnames(Typename)</code>
All field types	<code>Typename.types</code>

When a type is defined with an *inner* constructor, the default *outer* constructors are not available and have to be defined manually if need be. An inner constructor is best used to check whether the parameters conform to certain (invariance) conditions. Obviously, these invariants can be violated by accessing and modifying the fields directly, unless the type is defined as immutable. The `new` keyword may be used to create an object of the same type.

Type parameters are invariant, which means that `Point{Float64} <: Point{Real}` is false, even though `Float64 <: Real`. Tuple types, on the other hand, are covariant: `Tuple{Float64} <: Tuple{Real}`.

The type-inferred form of Julia's internal representation can be found with `code_typed()`. This is useful to identify where `Any` rather than type-specific native code is generated.

Missing and Nothing

Programmers Null	<code>nothing</code>
Missing Data	<code>missing</code>
Not a Number in Float	<code>NaN</code>
Filter missings	<code>collect(skipmissing([1, 2, missing])) == [1,2]</code>
Replace missings	<code>collect((df[:col], 1))</code>
Check if missing	<code>ismissing(x) not x == missing</code>

Exceptions

Throw SomeExcep	<code>throw(SomeExcep())</code>
Rethrow current exception	<code>rethrow()</code>
	<pre> struct NewExcep <: Exception v::String end </pre>
Define NewExcep	<code>Base.showerror(io::IO, e::NewExcep) = print(io, "A problem with \$(e.v)!")</code>
	<code>throw(NewExcep("x"))</code>
Throw error with msg text	<code>error(msg)</code>
	<pre> try # do something potentially iffy catch ex if isa(ex, SomeExcep) # handle SomeExcep elseif isa(ex, AnotherExcep) # handle AnotherExcep else # handle all others end finally # do this in any case end </pre>
Handler	

Modules

Modules are separate global variable workspaces that group together similar functionality.

Definition	<pre> module PackageName # add module definitions # use export to make definitions accessible end </pre>
Include filename.jl	<pre> include("filename.jl") </pre>
Load	<pre> using ModuleName # all exported names using ModuleName: x, y # only x, y import ModuleName # only ModuleName import ModuleName: x, y # only x, y import ModuleName.x, ModuleName.y # only x, y # Get an array of names exported by Module names(ModuleName) # include non-exports, deprecateds # and compiler-generated names names(ModuleName, all::Bool) #also show names explicitly imported from other modules names(ModuleName, all::Bool, imported::Bool) </pre>
Exports	

With `using Foo` you need to say `function Foo.bar(...` to extend module `Foo`'s function `bar` with a new method, but with `import Foo.bar`, you only need to say `function bar(...` and it automatically extends module `Foo`'s function `bar`.

Expressions

Julia is homoiconic: programs are represented as data structures of the language itself. In fact, everything is an expression `Expr`.

Symbols are **interned strings** prefixed with a colon. Symbols are more efficient and they are typically used as identifiers, keys (in dictionaries), or columns in data frames. Symbols cannot be concatenated.

Quoting `:(...)` or `quote ... end` creates an expression, just like `Meta.parse(str)`, and `Expr(:call, ...)`.

```

x = 1
line = "1 + $x"          # some code
expr = Meta.parse(line)  # make an Expr object
typeof(expr) == Expr    # true
dump(expr)              # generate abstract syntax tree
eval(expr) == 2         # evaluate Expr object: true

```

Macros

Macros allow generated code (i.e. expressions) to be included in a program.

Definition	<pre>macro macroname(expr) # do stuff end</pre>
Usage	<pre>@macroname(ex1, ex2, ...) or @macroname ex1 ex2 ... @test # equal (exact) @test x ≈ y # isapprox(x, y) @assert # assert (unit test) @which # types used @time # time and memory statistics @elapsed # time elapsed @allocated # memory allocated @profile # profile @spawn # run at some worker @spawnat # run at specified worker @async # asynchronous task @distributed # parallel for loop @everywhere # make available to workers</pre>
Built-in macros	

Rules for creating *hygienic* macros:

- Declare variables inside macro with `local`.
- Do not call `eval` inside macro.
- Escape interpolated expressions to avoid expansion: `$(esc(expr))`

Parallel Computing

Parallel computing tools are available in the `Distributed` standard library.

Launch REPL with N workers	<code>julia -p N</code>
Number of available workers	<code>nprocs()</code>
Add N workers	<code>addprocs(N)</code>
See all worker ids	<code>for pid in workers() println(pid) end</code>
Get id of executing worker	<code>myid()</code>
Remove worker	<code>rmprocs(pid)</code>
Run f with arguments args on pid	<code>r = remotecall(f, pid, args...) # or: r = @spawnat pid f(args) ... fetch(r)</code>
Run f with arguments args on pid (more efficient)	<code>remotecall_fetch(f, pid, args...)</code>
Run f with arguments args on any worker	<code>r = @spawn f(args) ... fetch(r)</code>
Run f with arguments args on all workers	<code>r = [@spawnat w f(args) for w in workers()] ... fetch(r)</code>
Make expr available to all workers	<code>@everywhere expr</code>
Parallel for loop with reducer function red	<code>sum = @distributed (red) for i in 1:10^6 # do parallelstuff end</code>
Apply f to all elements in collection coll	<code>pmap(f, coll)</code>

Workers are also known as concurrent/parallel processes.

Modules with parallel processing capabilities are best split into a functions file that contains all the functions and variables needed by all workers, and a driver file that handles the processing of data. The driver file obviously has to import the functions file.

A non-trivial (word count) example of a reducer function is provided by [Adam DeConinck](#).

I/O

Read stream	<pre>stream = stdin for line in eachline(stream) # do stuff end</pre>
Read file	<pre>open(filename) do file for line in eachline(file) # do stuff end end</pre>
Read CSV file	<pre>using CSV data = CSV.read(filename)</pre>
Write CSV file	<pre>using CSV CSV.write(filename, data)</pre>
Save Julia Object	<pre>using JLD save(filename, "object_key", object, ...)</pre>
Load Julia Object	<pre>using JLD d = load(filename) # Returns a dict of objects</pre>
Save HDF5	<pre>using HDF5 h5write(filename, "key", object)</pre>
Load HDF5	<pre>using HDF5 h5read(filename, "key")</pre>

DataFrames

For `dplyr`-like tools, see [DataFramesMeta.jl](#).

Read Stata, SPSS, etc.	StatFiles Package
Describe data frame	<code>describe(df)</code>
Make vector of column <code>col</code>	<code>v = df[:col]</code>
Sort by <code>col</code>	<code>sort!(df, [:col])</code>
Categorical <code>col</code>	<code>categorical!(df, [:col])</code>
List <code>col</code> levels	<code>levels(df[:col])</code>
All observations with <code>col==val</code>	<code>df[df[:col] .== val, :]</code>
Reshape from wide to long format	<code>stack(df, [1:n;])</code> <code>stack(df, [:col1, :col2, ...])</code> <code>melt(df, [:col1, :col2])</code>
Reshape from long to wide format	<code>unstack(df, :id, :val)</code>
Make <code>Nullable</code>	<code>allowmissing!(df)</code> or <code>allowmissing!(df, :col)</code>
Loop over Rows	<code>for r in eachrow(df)</code> # do stuff. # r is Struct with fields of col names. <code>end</code>
Loop over Columns	<code>for c in eachcol(df)</code> # do stuff. # c is tuple with name, then vector <code>end</code>
Apply func to groups	<code>by(df, :group_col, func)</code> <code>using Query</code> <code>query = @from r in df begin</code> <code>@where r.col1 > 40</code> <code>@select {new_name=r.col1, r.col2}</code> <code>@collect DataFrame # Default:</code> <code>iterator</code> <code>end</code>
Query	

Introspection and reflection

Type	<code>typeof(name)</code>
Type check	<code>isa(name, TypeName)</code>
List subtypes	<code>subtypes(TypeName)</code>
List supertype	<code>supertype(TypeName)</code>
Function methods	<code>methods(func)</code>
JIT bytecode	<code>code_llvm(expr)</code>
Assembly code	<code>code_native(expr)</code>

Noteworthy packages and projects

Many core packages are managed by communities with names of the form Julia[Topic].

Statistics	JuliaStats
Scientific Machine Learning	SciML (DifferentialEquations.jl)
Automatic differentiation	JuliaDiff
Numerical optimization	JuliaOpt
Plotting	JuliaPlots
Network (Graph) Analysis	JuliaGraphs
Web	JuliaWeb
Geo-Spatial	JuliaGeo
Machine Learning	JuliaML
	DataFrames # linear/logistic regression
	Distributions # Statistical distributions
Super-used Packages	Flux # Machine learning
	Gadfly # ggplot2-likeplotting
	LightGraphs # Network analysis
	TextAnalysis # NLP

Naming Conventions

The main convention in Julia is to avoid underscores unless they are required for legibility.

Variable names are in lower (or snake) case: `somevariable`.

Constants are in upper case: `SOMECONSTANT`.

Functions are in lower (or snake) case: `somefunction`.

Macros are in lower (or snake) case: `@somemacro`.

Type names are in initial-capital camel case: `SomeType`.

Julia files have the `.jl` extension.

For more information on Julia code style visit the manual: [style guide](#).

Performance tips

- Avoid global variables.
- Write `type-stable` code.
- Use immutable types where possible.
- Use `sizehint!` for large arrays.
- Free up memory for large arrays with `arr = nothing`.
- Access arrays along columns, because multi-dimensional arrays are stored in column-major order.
- Pre-allocate resultant data structures.
- Disable the garbage collector in real-time applications: `disable_gc()`.
- Avoid the splat `(...)` operator for keyword arguments.
- Use mutating APIs (i.e. functions with `!` to avoid copying data structures.
- Use array (element-wise) operations instead of list comprehensions.
- Avoid `try-catch` in (computation-intensive) loops.
- Avoid `Any` in collections.
- Avoid abstract types in collections.
- Avoid string interpolation in I/O.
- `Vectorizing` does not improve speed (unlike R, MATLAB or Python).
- Avoid `eval` at run-time.

IDEs, Editors and Plug-ins

- `Juno` (editor)
- `Jupyter` (online IJulia notebook)
- `Emacs Julia mode` (editor)
- `Pluto.jl` (online IJulia notebook)
- `vim Julia mode` (editor)
- `VS Code extension` (editor)

Resources

- [Official documentation](#) .
- [Learning Julia page](#).
- [Month of Julia](#)
- [Community standards](#) .
- [Julia: A fresh approach to numerical computing \(pdf\)](#)
- [Julia: A Fast Dynamic Language for Technical Computing \(pdf\)](#)

Videos

- [The 5th annual JuliaCon 2018](#)
- [The 4th annual JuliaCon 2017 \(Berkeley\)](#)
- [The 3rd annual JuliaCon 2016](#)
- [Getting Started with Julia by Leah Hanson](#)
- [Intro to Julia by Huda Nassar](#)
- [Introduction to Julia for Pythonistas by John Pearson](#)

Country flag icons made by [Freepik](#) from www.flaticon.com is licensed by [CC 3.0 BY](#).